# Let's Start with the Bootloader

## Introduction

When a computer powers on, a piece of code called bootloader is loaded and takes the control of the computer. Usually, the goal of the bootloader is loading the kernel of an operating system from the disk to the main memory and gives the kernel the control over the computer. The firmware of a computer is the program which loads the bootloader, in IBM-compatible computers the name of this firmware is BIOS (Basic Input/Output System) <sup>1</sup>.

There is a place in the hard disk called *boot sector*, it is the first sector of a hard disk <sup>2</sup>, BIOS loads the content of the boot sector as a first step of running the operating system. Loading the boot sector's content means that BIOS reads the content from hard disk and loads it into the main memory (RAM). This loaded content from the boot sector should be the bootloader, and once its loaded into the main memory, the processor will be able to execute it as any other code that we use in our computers. So, the last step performed by BIOS in booting process is giving the control to the bootloader to do whatever it wants.

Before getting started in writing 539kernel, we need to write the bootloader that is going to load 539kernel from the disk. In IBM-compatible PCs that uses BIOS to perform the booting process, the size of the bootloader is limited to 512 bytes and due to this limited size and the need of using low-level services, the bootloader is usually written in assembly language, also, because of this limited size, we cannot depend on BIOS to load 539kernel instead of the bootloader and that's because a kernel is usually bigger than 512 bytes, therefore, a small bootloader is loaded by BIOS in order to load the bigger piece of code which is the kernel. The reason of this limited size of the bootloader is because of the size of a sector in the hard disk itself. Each sector in the hard disk has the size of 512 bytes and as we have mentioned, BIOS is going to load the content of the first sector of hard disk, the boot sector, which, of course, as any other sector its size is 512 bytes. Beside the bootloader's limited size, it is going to run on an x86 operating mode known as real mode <sup>3</sup>, what we need to know about that for now is that real mode is a 16-bit environment, so, even if the working processor is a 64-bit processor, we can only use 16-bit features of the processor, such as the registers of size 16 bits.

The booting process is too specific to the computer architecture as we have seen and it may differs between one architecture and another. Some readers, especially Computer Science students may notice that the academic textbooks of operating systems don't mention the bootloader or discuss it.

 $<sup>^{1}</sup>$ Before the advent of UEFI.

 $<sup>^{2}</sup>$ As we will see later, a magnetic hard disk has multiple stacked *platters*, each platter is divided into multiple tracks and inside each track there are multiple sectors.

 $<sup>^3</sup>$ The concept of x86 operating modes and the real mode will be discussed in more details latter.

## Hard Disk Structure

A hard disk consists of multiple platters which are stacked together one above the other, have you ever seen a CD or a DVD? A platter has exactly the same shape, refer to Figure @fig:a-platter. The both surfaces (top and down) of a platter are covered by a magnetic layer which stores the data. For each surface of a platter there is a read/write head on it, and as you guessed, the role of this head is to read from a surface or write to it, a head is attached to an arm. Those arms move horizontally, back and forth, and because the other end of all of those arms are attached to the same physical part, they will be moved back and forth together to the same location at the same time. Figure @fig:platters-arms-heads shows how the platters, arms and read/write heads are assembled together.




Figure 1: (A) Shows a platter when we see it from the side. (B) Shows a platter when we see it from top/down.

A surface of a platter is divided into a number of tracks and each track is divided into a number of sectors. In Figure @fig:tracks-sectors you can see how tracks and sectors are organized on a surface, the gray circles that have the same center (cocenteric) are the tracks, a track consists of a smaller parts which called sectors. A sector is the smallest unit that holds data in hard disks and as you know from our previous discussion, the first sector in a hard disk is known as boot sector.




Figure 2: Shows how the parts of a hard disk are assembled together.

When a command is sent to the hard disk to write some data to the it or read from it, at least two mechanical moves <sup>4</sup> are performed. The first move is taken by the arms, they move back or forth in order to be upon the track that contains the data we would like to read, this operation is known as *seek* operation. So, *seek time* is the time needed to put a specific track under a read/write head. After finishing the seek operation, the read/write head will be on the right track but it is on a random sector <sup>5</sup>, to reach the sector that we would like to read from (or write to) the platter rotates until the read/write head becomes upon the required sector. The speed of rotation is measured by a unit known as *revolutions per minute* (RPM) and the needed time to reach the required sector is known as *rotational latency*. Finally, the data will be *transferred* from the hard disk to the main memory, the time needed to transfer a number of bits known as *transfer time*.

Let's assume as an example a hard disk that has 3 platters, which means it has 6 surfaces, arms and read/write head. When the operating system request from the hard disk to seek a specific track, for instance track 3, all 6 heads will seek the track 3 and when the seek operation ends, the 6 heads will point to the same physical position on all 6 surfaces, that is, the top head of the first platter and the bottom head of it is going to point to that same place, but the first one on the top which the second is on the bottom, and so on for the other 4 remaining heads, the collection of all these tracks that the heads point to at some point of time is called a *cylinder*.




Figure 3: Shows Tracks and Sectors on a platter's surface.

Now, based on what we know about how a hard disk works, can we imagine what happens inside the hard disk when BIOS loads a bootloader? First, the arms will seek the track number  $0^6$ , that is, the arms move back or forth until they reach

 $<sup>^4</sup>$ This fancy term  $mechanical\ moves$  means that some physical parts of hard disk moves physically.

<sup>&</sup>lt;sup>5</sup>Not exactly random, can you tell why?

 $<sup>^6\</sup>mathrm{I}$  didn't mention that previously, but yes, the bootloader resides in track 0.

the track O, then the platter rotates until the read/write head become upon the sector O, finally, the content of sector O is transferred to the main memory.

## **BIOS Services**

We are in a journey of writing an operating system kernel, which means that we are dealing with a little bit harsh environment! Do you remember all libraries that we are lucky to have when developing normal software (user-space software), well, none of them are available right now! And they will not be available until we decide to make them so and work hard to do that. Even the simple function printf of C is not available.

But that's fine, for our luck, in this environment, where there is too little available for us to write our code, BIOS provides us with a bunch of useful services that we can used in real mode, we can use these services in our bootloader to get things done.

BIOS services are like a group of functions in high-level languages that is provided by some library, each function does something useful and we deal with those functions as black boxes, we don't know what's inside these functions but we know what they do and how to use them. So, basically, BIOS provides us a library of functions and we are going to use some of these functions in our bootloader.

BIOS services are divided into categories, there are video services category, disk services category, keyboard services category and so on. Each category is identified by a unique number called *interrupt number*. In high-level world, we witnessed the same concept but with different mechanism, for example, C standard library provides us with many services (functions) such as input/output functions, string manipulation functions, mathematical functions and so on, these functions are categorized and each category is label by the *library name*, for example, all input/output functions can be found in stdio.h and so on. In BIOS, for example, the category of video services has the interrupt number 10h. As mentioned earlier, the letterh after a number means that this number represented in hexadecimal numbering system, or for short, a hexadecimal number. Here, 10h doesn't equal the decimal number 10. When we use hexadecimal number we use h as a postfix or 0x as prefix, so 10h and 0x10 are equivalents.

Inside each services category, there is a bunch of services, each one can do a specific thing and they are identified by a number. Continuing with C analogy, a service is a function labeled by a name (e.g. printf) and this function reside in a library (e.g. stdio.h) which is same as a category of services in BIOS. As we said, the interrupt number 10h represents the category of video services, and the service of printing a character on a screen is represented by the number 0Eh.

Interrupts is a fundamental concept in x86 architecture. What we need to know about them right now is that they are a way to call a specific code which is

registered to the interrupt number and calling an interrupt in assembly is really simple, the instruction int is used as the following: int 10h. That's it! We use the instruction int and gives it the interrupt number that represent the code we would like to call as an operand. In this example, we are calling the code of interrupt 10h which is, as we mentioned multiple time, the category of BIOS video services. When the CPU executes this instruction, BIOS will be called and based on the interrupt number it will know that we want to use one of available video services, but which one exactly!

In the previous example, we actually didn't tell BIOS which video service we would like to use and to do that we need to specify service number in ah register before calling the interrupt.

```
mov ah, OEh int 10h
```

That's it, all the BIOS services can be used in this exact way. First we need to know what is the interrupt number that the service belongs to, then, we need to know the number of the service itself, we put the service number in the register ah then we call the interrupt by its number by using int instruction.

The previous code calls the service of printing a character on a screen, but is it complete yet? Actually no, we didn't specify what is the character that we would like to print. We need something like parameters in high-level languages to pass additional information for BIOS to be able to do its job. Well, lucky us! the registers are here to the rescue.

When a BIOS service needs additional information, that is, parameters. It expects to find these information in a specific register. For example, the service OEh in interrupt 10h expects to find the character that the user wants to print in the register al, so, the register al is one of service OEh parameters. The following code requests from BIOS to print the character S on the screen:

```
mov ah, OEh
mov al, 'S'
int 10h
```

## A Little Bit More of x86 Assembly and NASM

We need to learn a couple more things about x86 assembly to be able to start. In NASM, each line in the source code has the following format.

```
label: instruction operands
```

The label is optional, the operands depend on x86 instruction in use, if it doesn't get any operand then we don't need to write any operand. To write comments on NASM we begin with semi-colon and write whatever we like after it as a comment and the rest of the source line will be considered as a part of the comment.

A label is a way to give an instruction or a group of instructions a meaningful name, then we can use this name in other places in the source code to refer to this instruction/group of instructions, we can use labels for example to call this group of instructions or to get the starting memory address of these instructions. Sometimes, we may use labels to make the code more readable.

We can say that a label is something like the name of a function or variable in C, as we know a variable name in C is a meaningful name that represents the memory address of the place in the main memory that contains the value of a variable, the same holds for a function name. Labels in NASM works in the same way, under the hood it represents a memory address. The colon in label is also optional.

```
print_character_S_with_BIOS:
    mov ah, OEh
    mov al, 'S'
    int 10h
```

You can see in the code above, we gave a meaningful name print\_character\_S\_with\_BIOS for the bunch of instructions that prints the character S on the screen. After defining this label in our source code, we can use it anywhere in the same source code to refer to this bunch of instructions.

```
call_video_service int 10h
```

This is another example of labels. This time we eliminated the optional colon in label's name and the label here points to only one instruction. Please note that extra whitespaces and new lines doesn't matter in NASM, so, the following is equivalent to the one above.

```
call_video_service
    int 10h

Consider the following code, what do you think it does?
print_character_S_with_BIOS:
    mov ah, OEh
    mov al, 'S'

call_video_service:
    int 10h
```

Still it prints S on the screen. Introducing labels in the source code doesn't change its flow, the code will be executed sequentially whether we used the labels or not. The sequence of execution will not be changed by merely using labels, if we need to change the sequence of execution we need to use other methods than labels. You already know one of these methods which is calling an interrupt. So, we can say that labels are more general than a variable name or function name in C. A label is a human-readable name for a memory location which can contain anything, code or data!

## Jump and Return Unconditionally

Let's start this section with a simple question. What happens when we call a function in C? Consider the following C code.

```
main()
{
    int result = sum( 5, 3 );
    printf( "%d\n", result );
}
```

Here, the function main called a function named sum, this function reside in a different region in memory and by calling it we are telling the processor to go to this different region of memory and execute what's inside it, the function sum is going to do its job, and after that, in some magical way, the processor is going to return to the original memory region where we called sum from and proceed the execution of the code that follows the calling of sum, in this case, the printf function. How does the processor know where to return after completing the execution of sum?

The function which call another is named *caller* while the function which is called by the caller named *callee*, in the above C code, the caller is the function main while the callee is the function sum.

## x86 Calling Convention

When a program is running, a copy of its machine code is loaded in the main memory, this machine code is a sequence of instructions which are understandable by the processor, these instructions are executed by the processor sequentially, that is, one after another in each cycle in the processor. This is known as von Neumann architecture where both code and data are stored in the same memory and the processor uses this memory to read the instructions that should be executed, and manipulate the data which is stored in the same memory. There is another well-known architecture called Harvard architecture where the code and data are stored in two different memories, x86 uses von Neumann architecture. However, When a processor starts a new instruction cycle, it fetches the next instruction that should be executed from the main memory and executes it <sup>7</sup>. Each memory location in the main memory is represented and referred to by a unique memory address, that means each instruction in the machine code of a loaded program has a unique memory address, consider the following hypothetical example of the memory addresses of each instruction in the previous C code, note that the memory addresses in this example are by no means accurate.

```
100 main() {
110    int result = sum(5, 3);
```

<sup>&</sup>lt;sup>7</sup>The instruction cycle is also called *fetch-decode-execute cycle*.

```
120  printf( "%d\n", result );
130 }
250 int sum( int firstNumber, int secondNumber ) {
260  return firstNumber + secondNumber;
270 }
```

The number on the left is the hypothetical memory address of the code line in the right, that means the function main starts from the memory address 100 and so on. Also, we can see that the callee sum resides in a far region of memory from the caller main.

Program Counter is a part of computer architecture which stores the memory address for the instruction that will be executed in the next instruction cycle of the processor. In x86, the program counter is a register known as instruction pointer and its name is IP in 16-bit and EIP in 32-bit.

When the above C code runs for the first time, the value of the instruction pointer will be 100, that is, the memory address of the starting point of main function. When the instruction cycle starts, it reads the value of the instruction pointer register IP/EIP which is 100, it fetches the instruction which is stored in the memory location 100 and executes it <sup>8</sup>, then the memory address of the next instruction 110 will be stored in the instruction pointer register for the next instruction cycle. When the processor finishes the execution of the instruction of the memory location 110, this time, the value of IP/EIP will be 250 instead of 120 because, you know, we are calling the function sum which resides in the memory location 250.

Each running program has a *stack* which is a region of the program's memory <sup>9</sup>, that is, a place in the memory that belongs to the program and can store data, we will examine the details of stack later, but what is important for us now is the following, when another function is called, in our case sum, the memory address of the next instruction of the callee main is *pushed* <sup>10</sup> into the stack, so the memory address 120 will be pushed into the stack before calling sum, this address is called *return address*. Now, assume that the processor is executing the instruction in the memory location 270, that is, finishing the execution of the callee sum, after that the processor will find the return address which is 120 in the stack, get it and put it in the register IP/EIP for the next instruction cycle <sup>11</sup>. So, this is the answer of our original question in the previous section "How does the processor know where to return after completing the execution of sum?".

<sup>&</sup>lt;sup>8</sup>For the simplicity of explanation, the details of *decoding* have been eliminated.

 $<sup>^9{</sup>m The\ stack}$  as a region of memory (x86 stack) is not same as the  $data\ structure\ stack$ , the former implements the latter.

<sup>&</sup>lt;sup>10</sup>Push means store something in a stack, this term is applicable for both x86 stack and the data structure stack, as we have said previously, x86 stack is an implementation of the stack data structure.

<sup>&</sup>lt;sup>11</sup>By the way, this is, partially, the cause of buffer overflow bugs.

#### The Instructions call and ret

The instruction call in assembly works exactly in the same way that we have explained in the previous section, it is used to call a code that resides in a given memory address. The instruction call pushes the return address into the stack and to return to the caller, the callee should use the instruction ret when it finishes. The instruction ret gets the return address from the stack <sup>12</sup> and use it to resume the execution of the caller. Consider the following example.

```
print_two_times:
    call print_character_S_with_BIOS
    call print_character_S_with_BIOS
    ret

print_character_S_with_BIOS:
    mov ah, OEh
    mov al, 'S'
    int 10h
```

You can see here that we have used the code sample print\_character\_S\_with\_BIOS to define something like C functions by using the instructions call and ret. It should be obvious that the code of print\_two\_times prints the character S two times, as we have said previously, a label represents a memory address and print\_character\_S\_with\_BIOS is a label, the operand of call is the memory address of the code that we wish to call, the instructions of print\_character\_S\_with\_BIOS will be executed sequentially until the processor reaches the instruction ret, at this point, the return address is obtained from the stack and the execution of the caller is resumed.

call performs an unconditional jump, that means processor will always call to the callee, without any condition, later in this chapter we will see an instruction that performs a conditional jump, which only calls the callee when some condition is satisfied, otherwise, the execution of the caller continues sequentially with flow change.

#### The One-Way Unconditional Jump

Like call, the instruction jmp jumps to the specified memory address, but unlike call, it doesn't store the return address in the stack which means ret cannot be used in the callee to resume the caller's execution. We use jmp when we want to jump to a code that we don't need to return from it, jmp has the same functionality of goto statement in C. Consider the following example.

```
print_character_S_with_BIOS:
    mov ah, OEh
```

 $<sup>^{12}</sup>$ Actually it pops the value since we are talking about stack here.

```
mov al, 'S'
   jmp call_video_service

print_character_A_with_BIOS:
   mov ah, OEh
   mov al, 'A'

call_video_service:
   int 10h
```

Can you guess what is the output of this code? it is S and the code of the label print\_character\_A\_with\_BIOS will never be executed because of the line jmp call\_video\_service. If we remove the line of jmp from this code sample, A will be printed on the screen instead of S. Another example which causes infinite loop.

```
infinite_loop:
    jmp infinite_loop
```

## Comparison and Conditional Jump

In x86 there is a special register called *FLAGS* register <sup>13</sup>. It is the *status register* which holds the current status of the processor. Each usable bit of this register has its own purpose and name, for example, the first bit (bit 0) of FLAGS register is known as *Carry Flag* (CF) and the seventh bit (bit 6) is known as *Zero Flag* (ZF).

Many x86 instructions use FLAGS register to store their result on, one of those instructions is cmp which can be used to compare two integers which are passed to it as operands, when the comparison finished the processor stores the its result in FLAGS register. The following line compares the value which reside in the register al with 5: cmp al, 5.

Now, let's say that we would like to jump to a piece of code only if the value of all equals 5, otherwise, the code of the caller continues without jumping. There are multiple instructions that perform *conditional* jump based on the result of cmp. One of these instructions is je which means jump if equal, that is, if the two operands of the cmp instruction equals each other, then jump to the specified code. Another conditional jump instruction is jne which means jump if not equal, there are other conditional jump instructions to handle the other cases. We can see that the conditional jump instructions have the same functionality of if statement in C. Consider the following example.

#### main:

```
cmp al, 5
```

 $<sup>^{13}</sup>$ In 32-bit x86 processors its name is EFLAGS and in 64-bit its name is RFLAGS.

```
je the_value_equals_5
; The rest of the code of `main` label
```

This example jumps to the code of the label the\_value\_equals\_5 if the value of the register al equals 5. In C, the above assembly example will be something like the following.

```
main()
{
    if ( register_al == 5 )
        the_value_equals_5();

// The rest of the code
}
```

Like jmp, but unlike call, conditional jump instructions don't push the return address into the stack, which means the callee can't use ret to return and resume caller's code, that is, the jump will be one way jump. We can also imitate while loop by using conditional jump instructions and cmp, the following example prints S five times by looping over the same bunch of code.

```
mov bx, 5
loop_start:
    cmp bx, 0
    je loop_end
    call print_character_S_with_BIOS
    dec bx
    jmp loop_start
loop_end:
    ; The code after loop
```

You should be familiar with the most of the code of this sample, first we assign the value 5 to the register bx  $^{14}$ , then we start the label loop\_start which the first thing it does is comparing the value of bx with 0, when bx equals 0 the code jumps to the label loop\_end which contains the code after the loop, that is, it means that the loop ended. When bx doesn't equal 0 the label print\_character\_S\_with\_BIOS will be called to print S and return to the caller loop\_start, after that the instruction dec is used to decrease 1 form its operand, that is bx = bx - 1, finally, the label loop\_start will be called again and the code repeats until the value of bx reaches to 0. The equivalent code in C is the following.

<sup>&</sup>lt;sup>14</sup>Can you tell why we used bx instead of ax? [Hint: review the code of print\_character\_S\_with\_BIOS.]

```
int bx = 5;
while ( bx != 0 )
{
    print_character_S_with_BIOS();
    bx--;
}
// The code after loop
```

## **Load String**

It is well-known that 1 byte equals 8 bits. Moreover, there are two size units in x86 other than a byte. The first one is known as a *word* which is 16 bits, that is, 2 bytes, and the second one is known as *doubleword* which is 32 bits, that is, 4 bytes. Some x86 instructions have multiple variants to deal with these different size units, while the functionality of an instruction is the same, the difference will be in the size of the data that a variant of instruction deals with. For example, the instruction lods has three variants lodsb which works a byte, lodsw which works with a word and loadsd which works with a doubleword.

To simplify the explanation, let's consider lodsb which works with a single byte, its functionality is too simple, it reads the value of the register si which is interpreted as a memory address by the instruction, then it transfers a byte from the content of that memory address to the register al, finally, it increments the value of si by 1 byte. The same holds for the other variants of lods, only the size of the data, the used registers and the increment size are different, the register which is used by lodsw is ax <sup>15</sup> and si is incremented by 2 bytes, while lodsd uses the register eax <sup>16</sup> and si is incremented by 4 bytes. <sup>17</sup>

## NASM's Pseudoinstructions

When you encounter the prefix <sup>18</sup> pseudo before a word, you should know that it describes something fake, false or not real <sup>19</sup>. NASM provides us a number

<sup>&</sup>lt;sup>15</sup>Because the size of ax is a word

 $<sup>^{16}\</sup>mathrm{Because}$  the size of eax is a doubleword.

<sup>&</sup>lt;sup>17</sup>As an exercise, try to figure out why are we explaining the instruction lodsb in this chapter, what is the relation between this instruction and the bootloader that we are going to write? Hint: Review the code of print\_character\_S\_with\_BIOS and how to print a character by using BIOS services. If you can't figure the answer out don't worry, you will get it soon.

<sup>&</sup>lt;sup>18</sup>In linguistics, which is the science that studies languages, a prefix is a word (actually a morpheme) that is attached in the beginning of another word and changes its meaning, for example, in **un**do, **un** is a prefix.

<sup>&</sup>lt;sup>19</sup>For example, in algorithm design which is a branch of computer science, the term **pseudo**code means a code that is written in a fake programming language. Another example is the word **pseudo**science: A statement is a pseudoscience when it is claimed to be a scientific fact, but in reality it is not, that is, it doesn't follow the scientific method.

of **pseudo**instructions, that is, they are not real x86 instructions, the processor doesn't understand them and they can't be used in other assemblers <sup>20</sup>, on the other hand, NASM understands those instructions and can translate them to something understandable by the processor. They are useful, and we are going to use them to make the writing of the bootloader easier.

#### **Declaring Initialized Data**

The concept of declaring something is well-known by the programmers, In C for example, when you declare a function, you are announcing that this function exists, it is there, it has a specific name and takes the declared number of parameters <sup>21</sup>. The same concept holds when you declare a variable, you are letting the rest of the code know that there exists a variable with a specific name and type. When we declare a variable, without assigning any value to it, we say that this variable is uninitialized, that is, no initial value has been assigned to this variable when it is declared, later on, a value will be assigned to the variable, but not as early of its declaration. In contrast, a variable is initialized when a value is assigned to it when it's declared.

The pseudoinstructions db, dw, dd, dd, dd, dd, and do helps us to initialize a memory location with some data, and with using labels when can mimic the concept of initialized variables in C. As an example, ;et's consider db which declares and initialize a byte of data, the second letter of db means bytes.

## db 'a'

The above example reserves a byte in the memory, this is the declaration step, then the character **a** will be stored on this reserved byte of the memory, which is the initialization step.

In the above example we have used comma to declare three bytes and store the values a, b and c respectively on them, also, on memory these values will be stored *contiguously*, that is, one after another, the memory location <sup>22</sup> of the value b will be right after the memory location of value a and the same rule for c. Since a, b and c are of the same type, a character, we can write the previous code as the following and it gives as the same result.

#### db 'abc'

Also, we can declare different types of data in the same source line, given the above code, let's say that we would like to store the number 0 after the character c, this can be achieved by simply using a comma.

 $<sup>^{20}</sup>$ Unless, of course, they are provided in the other assembler as pseudoinstructions.

<sup>&</sup>lt;sup>21</sup>It is important to note that *declaring* a function in C differs from *defining* a function, the following declares a function: int foo(); You can see that the code block (the implementation) of foo is not a part of the declaration, once the code block of the function is presented, we say this is the *definition* of the function.

<sup>&</sup>lt;sup>22</sup>Hence, the memory address.

```
db 'abc', 0
```

Now, to make this data accessible from other parts of the code, we can use a label to represent the starting memory address of this data. Consider the following example which define the label our\_variable, after that, we can use this label to refer to the initialized data.

```
our_variable db 'abc', 0
```

#### Repeating with times

To repeat some source line multiple times, we can use the pseudoinstruction times which takes the number of desired repetitions as first operand and the instruction that we would like to execute repeatedly as second operand. The following example prints S five times on the screen.

```
times 5 call print_character_S_with_BIOS
```

Not only normal x86 instructions can be used with times, also NASM's pseudoinstructions can be used with times. The following example reserves 100 bytes of the memory and fills them with 0.

```
times 100 db 0
```

#### NASM's Special Expressions

In programming languages, an expression is a part in the code that evaluates a value, for example, x + 1 is an expression, also, x == 5 is an expression. On the other hand, a statement is a part of the code that performs some actions, for example, in C, x = 15 \* y; is a statement that assigns the values of an expression to the variable x.

NASM has two special expressions, the first one is \$ which points to the beginning of the assembly position of the current source line. So, one way of implementing infinite loop is the following: jmp \$. The second special expression is \$\$ which points to the beginning of the current section of assembly code.

#### The Bootloader

As you have learned previously, the size of the bootloader should be 512 bytes, the firmware loads the bootloader in the memory address 07C0h, also, the firmware can only recognize the data in the first sector as a bootloader when that data finishes with the magic code AA55h. When 539kernel's bootloader starts, it shows two messages for the user, the first one is The Bootloader of 539kernel. and the second one The kernel is loading..., after that, it is going to read the disk to find 539kernel and loads it to memory, after loading 539kernel to memory, the bootloader gives the control to the kernel by jumping

to the start code of the kernel. Right now, 539kernel doesn't exist, we haven't write it yet, instead of loading 539kernel, the bootloader is going to load a code that prints Hello World!, From Simple Assembly 539kernel!. In this section, we are going to write two assembly files, the bootloader bootstrap.asm and simple\_kernel.asm which is the temporary replacement of 539kernel, also, Makefile which complies the source code that will be presented in this section.

#### Implementing the Bootloader

Till now, you have learned enough to understand the most of the bootloader that we are going to implement, however, some details have not been explained in this chapter and have been delayed to be explained later. The first couple lines of the bootloader is an example of not explained concepts, our bootloader source code starts with the following.

```
start:
   mov ax, 07C0h
   mov ds, ax
```

First, we define a label named start, there is no practical reason to define this label <sup>23</sup>, the only reason of defining it is the readability of the code, when someone else tries to read the code, it should be obvious for her that start is the starting point of executing the bootloader.

The job of next two lines is obvious, we are moving the hexadecimal number 07C0 to the register ax then we move the same value to the register ds through ax, note that we can't store the value 07C0 directly in ds by using mov as the following: mov ds, 07C0h, due to that, we have put the value on ax and then moved it to ds, so, our goal was to set the value 07C0 in the register ds, this restriction of not being able to store to ds directly is something that the processor architecture decides. Now, you may ask why we want the value 07C0 in the register ds, this is a story for another chapter, just take these two lines on faith, and you will learn later the purpose of them. Let's continue.

```
mov si, title_string
call message_string
mov si, message_string
call print_string
```

This block of code prints the two messages that we mentioned earlier, both of them are represented by a separate label title\_string and message\_string, you can see that we are calling the code of a label print\_string that we didn't define yet, its name indicates that it prints a *string* of characters, and you can infer that the function print\_string receives the memory address of the string

<sup>&</sup>lt;sup>23</sup>Such as jump to it for example.

that we would like to print as a parameter in the register si, the implementation of print\_string will be examined in a minute.

```
call load_kernel_from_disk
imp 0900h:0000
```

These two lines represent the most important part of any bootloader, first a function named <code>load\_kernel\_from\_disk</code> is called, we are going to define this function in a moment, as you can see from its name, it is going to load the code of the kernel from disk into the main memory and this is the first step that make the kernel able to take the control over the system. When this function finishes its job and returns, a jump is performed to the memory address <code>0900h:000</code>, but before discussing the purpose of the second line let's define the function <code>load\_kernel\_from\_disk</code>.

```
load_kernel_from_disk:
   mov ax, 0900h
   mov es, ax
```

These couple of lines, also, should be taken on faith. You can see, we are setting the value 0900h on the register es. Now, we move to the most important part of this function.

```
mov ah, 02h
mov al, 01h
mov ch, 0h
mov cl, 02h
mov dh, 0h
mov dl, 80h
mov bx, 0h
int 13h

jc kernel_load_error
```

This block of code *loads* the kernel from the disk into the memory and to do that it uses the BIOS Service 13h which provides services that perform operations of the disks such as reading and writing. The service number which is 02h is specified on the register ah, this service read sectors from disks and loads them into the memory, the following registers should be set when we use the service 02h in 13h:

The value of register al is the number of sectors that we would like to read, in our case, we read only 1 sector, the size of our temporary kernel simple\_kernel.asm doesn't exceed 512 bytes. Please keep in mind that we are going to store our kernel right after the bootloader in the disk, knowing that, you can make sense of the registers' values ch, cl and dh that we will be explained next. The value of register ch is the track number we would like to read from, in our case, it is

the track 0. The values of the register c1 is the sector number that we would like to read its content, in our case, it is the second sector. The value of the register dh is the number of head. The values of of d1 specifies which disk we would like to read from, the value 0h in this register means that we would like to read the sector from a floppy disk, while the value 80h means we would like to read from the hard disk #0 and 81h for hard disk #1, in our case, the kernel is stored in the hard disk #0, so, the value of d1 should be 80h. Finally, the value of the register bx is the memory address that the content will be loaded to, in our case, we are reading one sector, and its content will be stored on the memory address 0h <sup>24</sup>.

When the content is loaded successfully, the BIOS Service 13h:02h is going to set the carry flag to 0, otherwise, it sets the carry flag to 1 and stores the error code in register ax, the instruction jc is a conditional jump instruction that jumps when CF = 1, that is, when the value of the carry flag is 1. That means our bootloader is going to jump to the label kernel\_load\_error when the kernel isn't loaded correctly.

If the kernel is loaded correctly, the function <code>load\_kernel\_from\_disk</code> returns by using the instruction <code>ret</code> which makes the processor to resume the main code of our bootloader and executes that instruction which is after <code>callload\_kernel\_from\_disk</code>, this instruction is instruction <code>jmp</code> <code>0900h:0000</code> which gives the control to the kernel by jumping to its starting point, that is, the memory location where we loaded our kernel in. This time, the operand of <code>jmp</code> is an <code>explicit</code> memory address <code>0900h:0000</code>, it has two parts, the first part is the one before the colon, you can see that it is the same value that we have loaded in the register <code>es</code> in the beginning of <code>load\_kernel\_from\_disk</code> function. The second part of the memory address is the one after the colon, it is <code>Oh</code> <sup>25</sup> which is the <code>offset</code> that we have specified in the register <code>bx</code> in <code>load\_kernel\_from\_disk</code> before calling <code>O2h:13h</code>, the both parts combined represent the memory address that we have loaded our kernel into and the details of the two parts of this memory address will be discussed in chapter .

Now we have finished the basic code of the bootloader, we can start defining that labels that we have used before in its code. We start with the label kernel\_load\_error which simply prints an error message that is stored on another label, as we have done previously, the label kernel\_load\_error prints the error message by using the function print\_string, after printing the message, nothing can be done, so, kernel\_load\_error enters an infinite loop.

```
kernel_load_error:
   mov si, load_error_string
   call print_string

imp $
```

 $<sup>^{24}\</sup>mathrm{Not}$  exactly the memory address 0h. We will see why in the next chapter.

 $<sup>^{25}\</sup>mathrm{Here},\,\mathtt{0h}$  is equivalent to  $\mathtt{0000}.$ 

Our previous samples of using the BIOS Service OEh:10h were printing only one character, in real world, we need to print a *string* of characters and that's what the function print\_string exactly does, it takes the memory address which is stored in the register si and prints the character which is stored in it, then it goes to the next memory address and prints the character which is stored in it and so on, that is, print\_string prints a string character by character by using loop. So, you may ask, how print\_string can know when should it stop? There should be a stop condition in any loop, otherwise, it is an infinite loop.

A string in C programming language, as in our situation, is an array of characters, and the same problem of "where does a string end" is encountered in C programming language, to solve the problem, each string in C programming language ends with a special character named null character and represented by the symbol 0 in C  $^{26}$ , so, you can handle any string in C character by character and once you encounter the null character 0 that means you have reached the end of the string. We are going to use the same mechanism in our bootloader to recognize the end of a string by putting the value 0 as a marker in the end of the string. By using this way, we can now use the service 0Eh: 10h to print any string character by character through a loop and once we encounter the value 0 we can stop the printing.

```
print_string:
    mov ah, OEh
print_char:
    lodsb
    cmp al, 0
    je printing_finished
    int 10h
    jmp print_char
printing_finished:
    mov al, 10d; Print new line
    int 10h
    ; Reading current cursor position
    mov ah, 03h
    mov bh, 0
    int 10h
    ; Move the cursor to the beginning
    mov ah, 02h
```

<sup>&</sup>lt;sup>26</sup>This type of strings named null-terminated strings.

```
mov dl, 0 int 10h
```

ret

When print\_string starts, the BIOS services number OEh is loaded in ah, this operation need to take a place just one time for each call for print\_string, so it is not a part of the next label print\_char which is a part of print\_string and it will be executed after moving OEh to ah.

As you can remember, that parameter of print\_string is the memory address which contains the string that we would like to print, this parameter is passed to print\_string via the register si, so, the first thing print\_char does is using the instruction lodsb which is going to transfer the first character of the string to the register al and increase the value of si by 1 byte, after that, we check the character that has been transferred from the memory, if it is 0, that means we have reached to the end of the string and the code jumps to the label printing\_finished, otherwise, the interrupt 10h of BIOS is called to print the content of the register al on the screen, then we jump to print\_char again to repeat this operation until we read the end of the string. When printing a string finishes, the label printing\_finished starts by printing a new line after the string, the new line is represented by the number 10 in ASCII after that we are going to use the service 03h to read the current position of the cursor, then we use the service 02h to set the cursor to position 0 by passing it to the register dl, otherwise, the messages in the new lines will be printed in the position where the previous string finished, finally the function returns to the caller by using the instruction ret.

```
title_string db 'The Bootloader of 539kernel.', 0
message_string db 'The kernel is loading...', 0
load_error_string db 'The kernel cannot be loaded', 0
```

The code above defines the strings that have been used previously in the source code, note the last part of each string which is the null character that indicates the end of a string <sup>27</sup>.

Now, we have written our bootloader and the last thing to do is to put the magic code in the end of it, the magic code which is a 2 bytes value should be the last two bytes in the first sector, that is, in the locations 510 and 511  $^{28}$ , otherwise, the firmware will not recognize the content of the sector as a bootloader. To ensure that the magic code is written on the correct location, we are going to fill the empty space between the last part of bootloader code and the magic code by zeros, this can be achieved by this line.

```
times 510-($-$$) db 0
```

 $<sup>^{27}</sup>$ Exercise: What will be the behavior of the bootloader if we remove the null character from title\_string and message\_string and keeps it in load\_error\_string?

<sup>&</sup>lt;sup>28</sup>The location number starts from 0.

So, the instruction db will be called 510-(\$-\$\$) times, this expression gives us the remaining empty space in our bootloader before the magic code, and because the magic code is a 2 bytes value we subtract (\$-\$\$) from 510 instead of 512, we will use these two bytes for the magic code, the expression (\$-\$\$) uses the special expressions of NASM \$ and \$\$\$ and it gives the size of the bootloader code until this line. Finally, the magic code is presented.

```
dw 0xAA55
```

## Implementing simple\_kernel.asm and Makefile

The simple\_kernel.asm which the bootloader loads is too simple, it prints the message "Hello World!, From Simple Assembly 539kernel!", we don't need to go through its code in details since you know most of it.

```
start:
    mov ax, cs
    mov ds, ax
    ; --- ;
    mov si, hello_string
    call print_string
    jmp $
print_string:
    mov ah, OEh
print char:
    lodsb
    cmp al, 0
    je done
    int 10h
    jmp print_char
done:
    ret
```

hello\_string db 'Hello World!, From Simple Assembly 539kernel!', 0

The only lines that you are not familiar with until now are the first two lines in the label start which will be explained in details in the next chapter. Finally